

Table of Contents

ANSYS APDL Emniyet Kemer Analizi	3
1. Tam APDL Kodu	3
2. Komut Satırı Açıklamaları	5
1. Genel Ayarlar	5
2. Ön İşlemci (Preprocessor) ve Geometri Oluşturma	5
3. Boolean İşlemleri ve Radyus (Fillet) Atma	5
4. Eleman Tipi, Malzeme ve Ağ (Mesh) Yapısı	6
5. Çözüm (Solution) Modülü ve Sınır Koşulları	6
6. Ard İşlemci (Postprocessor) ve Sonuçların Çıkarılması	7

ANSYS APDL Emniyet Kemer Analizi

Determine the stress and deformation of the prototype seatbelt component shown in the figure below if it is subjected to tensile load of 1000 lbf.



The seatbelt component is made of steel, has an overall length of about 2.5 inches and is $3/32'' = 0.09375$ inches thick. For simplicity we will analyze only the right, or 'tongue' portion of 1.50 inches. The part is as shown in Figure 2.



Figure 2. Seatbelt "tongue" is shown in purple (top half) with dimensions!

- I. Calculate the deformation in UX shown above,
- II. Calculate the Equivalent Stress and find out where it occurs,

1. Tam APDL Kodu

Aşağıdaki kod, GUI (arayüz) kullanılmadan, parametrik modelleme (top-down) mantığıyla hazırlanmıştır.

```
/FILNAM, Seatbelt
/TITLE, Seatbelt Geometry

/PREP7
! Geometri ve parametrelerin tanımlanması
thickness = 0.09375
fillet = 0.03125

RECTNG, 0, 1.5, 0, 0.75
RECTNG, 0, 0.375, 0, 0.40625

! Yuvanın (slot) doğru konuma taşınması
AGEN, , 2, , , 0.75, , , , 1
PCIRC, 0.35, , 0, 90,

AOVLAP, ALL
ADELE, 4, , , 1
ADELE, 2, , , 1

! Alanları silip çizgileri (line) bırakıyoruz, fillet atamak için
ADEL, ALL
LFILLT, 3, 2, 0.25, ,
LFILLT, 7, 8, 0.0625, ,
```

```
LFILLT, 7, 6, fillet, ,

AL, ALL

! Eleman tipi (Quad 8 node 183) ve Real Constant ataması
ET, 1, PLANE183
KEYOPT, 1, 3, 3 ! Plane stress w/ thickness
R, 1, thickness

! Malzeme özellikleri (Çelik)
MP, EX, 1, 30.0E06
MP, NUXY, 1, 0.3

! Mesh işlemi
LESIZE, 5, , , 8, , , , 1
AESIZE, ALL, 0.02
AMESH, ALL
FINISH

/SOLU
! DİKKAT: LFILLT ve AOV LAP sonrası çizgi numaraları değişir.
! L PLOT ve /PNUM, LINE, 1 ile mutlaka TEYİT ETMELİSİN.

! Simetri sınır koşulları (Alt kenar - Y=0 eksenini)
DL, 16, , SYMM
DL, 14, , SYMM

! Sol kenar ankastre (UX = 0)
DL, 15, , UX, 0

! Basıncı yükünün hesabı ve uygulanması (Alan çizgisine uygulanıyor)
basinc = 1000 / (thickness * 0.75)
SFL, 6, PRES, basinc

ALLSEL, ALL
SOLVE
FINISH

/POST1
! Plot çıktısı için arkası beyaz olacak şekilde renk ayarları
/RGB, INDEX, 100, 100, 100, 0
/RGB, INDEX, 80, 80, 80, 13
/RGB, INDEX, 60, 60, 60, 14
/RGB, INDEX, 0, 0, 0, 15

! Simetriyi genişleterek tam modeli gösterme
/EXPAND, 2, RECT, HALF, , 0.0001

! UX Deformasyon kontur grafiği
```

```
PLNSOL, U, X, 2, 1.0
```

```
! Eşdeğer (Von Mises) Gerilme kontur grafiği  
PLNSOL, S, EQV, 2
```

```
/REPLLOT  
FINISH
```

2. Komut Satırı Açıklamaları

Aşağıda, hazırladığımız APDL kodunun her bir komutunun ne anlama geldiği bölümler halinde açıklanmıştır.

1. Genel Ayarlar

`/FILNAM, Seatbelt` : Oluşturulacak veritabanı (database) ve sonuç dosyalarının adını "Seatbelt" olarak ayarlar (örn. `Seatbelt.db`).

`/TITLE, Seatbelt Geometry` : Ekranın sol alt köşesinde ve alınacak grafik çıktılarında (plot) görünecek başlık metnini belirler.

2. Ön İşlemci (Preprocessor) ve Geometri Oluşturma

`/PREP7` : ANSYS'in ön işlemci (Preprocessor) modülüne girer. Geometri oluşturma, malzeme tanımlama ve mesh atma işlemleri sadece bu modülde yapılabilir.

`thickness = 0.09375` ve `fillet = 0.03125` : Bunlar APDL içinde tanımlanan parametrik değişkenlerdir. İleride formüllerde veya komutlarda uzun uzun sayı yazmak yerine bu isimler kullanılır.

`RECTNG, 0, 1.5, 0, 0.75` : (X1, X2, Y1, Y2) koordinatlarına sahip bir dikdörtgen alan (Area) oluşturur. Bu, emniyet kemeri dilinin ana gövdesinin üst yarısıdır.

`RECTNG, 0, 0.375, 0, 0.40625` : Ortadaki kilit yuvası (slot) için kullanılacak olan dikdörtgen alanı oluşturur. (*Not: Bu başlangıçta orijinde oluşturulup sonraki komutla taşınacaktır.*)

`AGEN, , 2, , , 0.75, , , , 1` : Alan kopyalama/taşıma (Area Generate) komutudur. 2 numaralı alanı alır, X, Y, Z yönlerinde ne kadar taşınacağını belirtir (burada sadece Y yönünde 0.75 birim taşınmış).

`PCIRC, 0.35, , 0, 90` : Kısmi veya tam dairesel alan oluşturur. 0.35 yarıçapında, 0 ile 90 derece arasında (çeyrek daire) bir alan yaratır. Sol taraftaki oyuğu temsil eder.

3. Boolean İşlemleri ve Radyus (Fillet) Atma

`A0VLAP, ALL` : (Area Overlap) Kesişen tüm alanları üst üste bindirir, kesiştikleri sınırlardan böler ve yeni alanlar/çizgiler yaratır. Boolean işlemidir.

ADELE, 4, , , 1 ve ADELE, 2, , , 1 : (Area Delete) Belirtilen numaralı alanları siler. Kilit yuvası (slot) ve sol kenardaki daire parçasını ana gövdeden çıkarıp “delik” açmak için iç kısımları siler.

ADEL, ALL : (ÖNEMLİ TRICK) Tüm alanların *yüzeyini* siler ancak sınır çizgilerini (Lines) ve noktalarını (Keypoints) ekranda bırakır. Bu işlem, alanların köşelerine doğrudan komutla radyus (fillet) atabilmek için yapılır.

LFILLT, 3, 2, 0.25, , : İki kesişen çizgi arasına Line Fillet (Radyus) atar. 3 ve 2 numaralı çizgiler arasına 0.25 inçlik kavisli bir çizgi oluşturur. (Dış sağ üst köşe)

LFILLT, 7, 8, 0.0625, , ve LFILLT, 7, 6, fillet, , : İç kilit yuvasının (slot) köşelerine radyus atar. fillet kelimesi en başta tanımladığımız 0.03125 değişkenini çeker.

AL, ALL : (Area from Lines) Ekranda kalan tüm dış ve iç sınır çizgilerini (Lines) kullanarak boşlukları doldurur ve nihai, delikli, kavisli tek bir kapalı Alan (Area) oluşturur.

4. Eleman Tipi, Malzeme ve Ağ (Mesh) Yapısı

ET, 1, PLANE183 : 1 numaralı eleman referansına PLANE183 tipini atar. Bu, 2 boyutlu, 8 noktalı (quadrilateral) yüksek hassasiyetli bir elemandır.

KEYOPT, 1, 3, 3 : 1 numaralı eleman (PLANE183) için özel ayar (Key Option) yapar. 3. hanedeki “3” değeri, bu elemanın “Plane Stress with Thickness” (Kalınlıklı Düzlem Gerilmesi) özelliği göstereceğini belirtir.

R, 1, thickness : (Real Constant) 1 numaralı sabit setini oluşturur ve içeriğine parça kalınlığını (thickness = 0.09375) atar. KEYOPT ayarı sayesinde ANSYS 2D hesap yaparken bu kalınlığı formüllere dahil eder.

MP, EX, 1, 30.0E06 : 1 numaralı malzeme için Elastisite (Young) Modülünü (EX) atar. Çelik için 30×10^6 Psi.

MP, NUXY, 1, 0.3 : 1 numaralı malzeme için Poisson oranını (NUXY) 0.3 olarak atar.

LESIZE, 5, , , 8, , , , 1 : (Line Element Size) Sadece 5 numaralı çizgi üzerine toplam 8 adet mesh elemanı yerleşmesini zorunlu kılar. Bölgesel mesh kontrolüdür.

AESIZE, ALL, 0.02 : (Area Element Size) Geri kalan tüm alanlardaki genel mesh elemanlarının maksimum kenar boyutunu 0.02 inç olarak ayarlar.

AMESH, ALL : Bütün ayarlamaları bitirdikten sonra tüm alana (Area) üçgen/dörtgen ağı örür (Meshing).

FINISH : Ön işlemci modülünü (PREP7) kapatır.

5. Çözüm (Solution) Modülü ve Sınır Koşulları

/SOLU : Çözüm (Solution) modülüne girer. Sınır koşulları (mesnetler) ve kuvvetler burada uygulanır, denklem burada çözülür.

DL, 16, , SYMM ve DL, 14, , SYMM : (Displacement on Line) Çizgiler üzerine simetri (SYMM) mesneti tanımlar. Modelin sadece üst yarısını modellediğimiz için, alt kesik kenardaki (Y=0 eksenindeki) çizgilere bu uygulanır. *Not: Çizgi numaraları değişebilir.*

DL, 15, , UX, 0 : 15 numaralı çizginin sadece X yönünde (UX) hareketini engeller (Değeri 0 yapar). Emniyet kemeri tokasının sol taraftan tutulduğunu simüle eder.

basinc = 1000 / (thickness * 0.75) : Basınç değerini hesaplar. 1000 lbf yük, kalınlık * kilit yuvası yüksekliği kadar bir alana dağıtılacaktır ($P = F / A$).

SFL, 6, PRES, basinc : (Surface Load on Line) 6 numaralı çizgi yüzeyine hesaplanan basıncı (PRES) uygular. Kuvvet sağa doğru olduğu için pozitif değer verilir.

ALLSEL, ALL : Modeldeki tüm düğüm noktalarını, çizgileri, elemanları seçili hale getirir. İşlemlerden önce güvenlik amaçlıdır, hiçbir noktanın çözüm dışı kalmamasını sağlar.

SOLVE : Kurulan matrisleri çözer ve analizi tamamlar. (Solution is done!)

FINISH : Çözüm modülünü kapatır.

6. Ard İşlemci (Postprocessor) ve Sonuçların Çıkarılması

/POST1 : Genel sonuç inceleme modülüne girer.

/RGB, INDEX, ... komutları: Normalde ANSYS arka planı siyahtır. Beyaz arkaplan çıktısı almak için indeks renk haritasındaki siyah ve beyazın yerlerini matematiksel olarak değiştirir.

/EXPAND, 2, RECT, HALF, , 0.0001 : Görsel bir hiledir. Sadece üst yarısını çözdüğümüz modeli, X ekseninde aşağıya doğru aynalayarak ekranda "tam (bütün)" bir parça gibi görünmesini sağlar. Sonuç raporlarında daha şık durur.

PLNSOL, U, X, 2, 1.0 : (Plot Nodal Solution) Düğüm noktalarındaki yer değiştirme (Deformation - U) değerlerinin sadece X eksenindeki (X) bileşenini renkli kontur grafiği olarak çizer.

PLNSOL, S, EQV, 2 : Düğüm noktalarındaki gerilmelerin (Stress - S) Von Mises Eşdeğer Gerilmesini (EQV) renkli kontur grafiği olarak çizer.

/REPLLOT : Yapılan renk ve simetri güncellemelerinin ekrana yansımaları için grafiği yeniler.

FINISH : Ard işlemciden çıkar.

The information in this document is cited from [UCH Wiki](https://wiki.ulascemh.com/).

From:
<https://wiki.ulascemh.com/> - UCH

Permanent link:
<https://wiki.ulascemh.com/doku.php?id=en:eng:cada:fea:apdl:example:seatbelt&rev=1775153659>

Last update: 2026/04/02 18:14

